contract PortionStaking is Ownable {
    using SafeMath for uint256;
    using SafeMath for uint16;

    /**
     *
     * @dev User reflects the info of each user
     *
     *
     * @param {total_invested} how many tokens the user staked
     * @param {total_withdrawn} how many tokens withdrawn so far
     * @param {lastPayout} time at which last claim was done
     * @param {depositTime} Time of last deposit
     * @param {totalClaimed} Total claimed by the user
     *
     */
    struct User {
        uint256 total_invested;
        uint256 total_withdrawn;
        uint256 lastPayout;
        uint256 depositTime;
        uint256 totalClaimed;
    }

    /**
     *
     * @dev PoolInfo reflects the info of each pools
     *
     * To improve precision, we provide APY with an additional zero. So if APY is 12%, we provide
     * 120 as input.lockPeriodInDays would be the number of days which the claim is locked. So if we want to
     * lock claim for 1 month, lockPeriodInDays would be 30.
     *
     * @param {apy} Percentage of yield produced by the pool
     * @param {lockPeriodInDays} Amount of time claim will be locked
     * @param {totalDeposit} Total deposit in the pool
     * @param {startDate} starting time of pool
     * @param {endDate} ending time of pool in unix timestamp
     * @param {minContrib} Minimum amount to be staked
     *
     */

    struct Pool {
        uint16 apy;
        uint16 lockPeriodInDays;
        uint256 totalDeposit;
        uint256 startDate;
        uint256 endDate;
        uint256 minContrib;
    }

    IERC20 public token; //Token address
    uint8 public depositFeePercent; //Percentage of fee deducted (/1000)
    uint8 public withdrawFeePercent;

    mapping(uint256 => mapping(address => User)) public users;

    Pool[] public poolInfo;

    event Stake(address indexed addr, uint256 amount);
    event Claim(address indexed addr, uint256 amount);

    constructor() {
        token = IERC20(0x6D0F5149c502faf215C89ab306ec3E50b15e2892);
        depositFeePercent = 10; //1% deposit fee
        withdrawFeePercent = 20;// 2% deposit fee
    }

    receive() external payable {
        revert("BNB deposit not supported");
    }

    /**
     *
     * @dev get length of the pools
     *
     * @return {uint256} length of the pools
     *
     */
    function poolLength() public view returns (uint256) {
        return poolInfo.length;
    }

    /**
     *
     * @dev get info of all pools
     *
     * @return {PoolInfo[]} Pool info struct
     *
     */
    function getPools() internal view returns (Pool[] memory) {
        return poolInfo;
    }

    /**
     *
     * @dev add new period to the pool, only available for owner
     *
     */
    function add(
        uint16 _apy,
        uint16 _lockPeriodInDays,
        uint256 _endDate,
        uint256 _minContrib
    ) public onlyOwner {
        poolInfo.push(
            Pool({
                apy: _apy,
                lockPeriodInDays: _lockPeriodInDays,
                totalDeposit: 0,
                startDate: block.timestamp,
                endDate: _endDate,
                minContrib: _minContrib
            })
        );
    }

    /**
     *
     * @dev update the given pool's Info
     *
     */
    function set(
        uint256 _pid,
        uint16 _apy,
        uint16 _lockPeriodInDays,
        uint256 _endDate,
        uint256 _minContrib
    ) public onlyOwner {
        require(_pid < poolLength(), "Invalid pool Id");

        poolInfo[_pid].apy = _apy;
        poolInfo[_pid].lockPeriodInDays = _lockPeriodInDays;
        poolInfo[_pid].endDate = _endDate;
        poolInfo[_pid].minContrib = _minContrib;
    }

    /**
     *
     * @dev depsoit tokens to staking for  allocation
     *
     * @param {_pid} Id of the pool
     * @param {_amount} Amount to be staked
     *
     * @return {bool} Status of stake
     *
     */
    function stake(uint8 _pid, uint256 _amount) external returns (bool) {
        Pool memory pool = poolInfo[_pid];

        require(_amount >= pool.minContrib, "Invalid amount!");

        require(
            token.allowance(msg.sender, address(this)) >= _amount,
            "Staking : Set allowance first!"
        );

        bool success = token.transferFrom(msg.sender, address(this), _amount);
        require(success, "Staking : Transfer failed");

        uint256 fee = _amount.mul(depositFeePercent).div(1000);
        token.transfer(owner(), fee);
        _amount = _amount.sub(fee);

        _stake(_pid, msg.sender, _amount);

        return success;
    }

    function _stake(
        uint8 _pid,
        address _sender,
        uint256 _amount
    ) internal {
        User storage user = users[_pid][_sender];
        Pool storage pool = poolInfo[_pid];

        uint256 stopDepo = pool.endDate.sub(pool.lockPeriodInDays.mul(1 days));

        require(
            block.timestamp <= stopDepo,
            "Staking is disabled for this pool"
        );

        user.total_invested = user.total_invested.add(_amount);
        pool.totalDeposit = pool.totalDeposit.add(_amount);

        user.lastPayout = block.timestamp;
        user.depositTime = block.timestamp;

        emit Stake(_sender, _amount);
    }

    /**
     *
     * @dev claim accumulated  reward for a single pool
     *
     * @param {_pid} pool identifier
     *
     * @return {bool} status of claim
     */

    function claim(uint8 _pid) public returns (bool) {
        _claim(_pid, msg.sender);

        return true;
    }

    /**
     *
     * @dev claim accumulated  reward from all pools
     *
     * Beware of gas fee!
     *
     */
    function claimAll() public returns (bool) {
        uint256 length = poolInfo.length;

        for (uint8 pid = 0; pid < length; ++pid) {
            _claim(pid, msg.sender);
        }

        return true;
    }

    /**
     *
     * @dev check whether user can claim or not
     *
     * @param {_pid}  id of the pool
     * @param {_addr} address of the user
     *
     * @return {bool} Status of claim
     *
     */

    function canClaim(uint8 _pid, address _addr) public view returns (bool) {
        User storage user = users[_pid][_addr];
        Pool storage pool = poolInfo[_pid];

        return (block.timestamp >=
            user.depositTime.add(pool.lockPeriodInDays.mul(1 days)));
    }

    /**
     *
     * @dev withdraw tokens from Staking
     *
     * @param {_pid} id of the pool
     * @param {_amount} amount to be unstaked
     *
     * @return {bool} Status of stake
     *
     */
    function unStake(uint8 _pid, uint256 _amount) external returns (bool) {
        User storage user = users[_pid][msg.sender];
        Pool storage pool = poolInfo[_pid];

        require(user.total_invested >= _amount, "You don't have enough funds");

        require(canClaim(_pid, msg.sender), "Stake still in locked state");

        _claim(_pid, msg.sender);

        pool.totalDeposit = pool.totalDeposit.sub(_amount);
        user.total_invested = user.total_invested.sub(_amount);

        uint256 fee = _amount.mul(withdrawFeePercent).div(1000);
        token.transfer(owner(), fee);
        _amount = _amount.sub(fee);

        safeTransfer(msg.sender, _amount);

        return true;
    }

    function _claim(uint8 _pid, address _addr) internal {
        User storage user = users[_pid][_addr];

        uint256 amount = _payout(_pid, _addr);

        if (amount > 0) {
            user.total_withdrawn = user.total_withdrawn.add(amount);

            safeTransfer(_addr, amount);

            user.lastPayout = block.timestamp;

            user.totalClaimed = user.totalClaimed.add(amount);
        }

        emit Claim(_addr, amount);
    }

    function _payout(uint8 _pid, address _addr)
        public
        view
        returns (uint256 value)
    {
        User storage user = users[_pid][_addr];
        Pool storage pool = poolInfo[_pid];

        uint256 from = user.lastPayout > user.depositTime
            ? user.lastPayout
            : user.depositTime;
        uint256 to = block.timestamp > pool.endDate
            ? pool.endDate
            : block.timestamp;

        if (from < to) {
            value = value.add(
                user.total_invested.mul(to.sub(from)).mul(pool.apy).div(
                    365 days * 1000
                )
            );
        }

        return value;
    }

    /**
     *
     * @dev safe  transfer function, require to have enough  to transfer
     *
     */
    function safeTransfer(address _to, uint256 _amount) internal {
        uint256 Bal = token.balanceOf(address(this));
        if (_amount > Bal) {
            token.transfer(_to, Bal);
        } else {
            token.transfer(_to, _amount);
        }
    }

    /**
     *
     * @dev update fee values
     *
     */
    function updateFeeValues(uint8 _depositFee, uint8 _withdrawFee)
        public
        onlyOwner
    {
        depositFeePercent = _depositFee;
        withdrawFeePercent = _withdrawFee;
    }
}
function updateFeeValues(uint8 _depositFee, uint8 _withdrawFee)
    public
    onlyOwner
{
    depositFeePercent = _depositFee;
    withdrawFeePercent = _withdrawFee;
}

/**
 *
 * @dev update fee values
 *
 */